

Maxima, minima, suprema a infima

Maxima a minima konečných množin

Maximum nebo minimum konečného počtu čísel se zjistí příkazy `Max`, `Min`.

```
Max[6, -1, 4, -2]
```

```
6
```

```
Min[6, -1, 4, -2]
```

```
-2
```

Maxima a minima, suprema a infima nekonečných množin

Uvedené příkazy nepracují na nekonečných množinách.

```
Max[1 / n, 1 < n < Infinity]
```

$$\max\left(\frac{1}{n}, 1 < n < \infty\right)$$

```
Min[Table[1 / n, {n, 2, 100}]]
```

$$\frac{1}{100}$$

```
Min[Table[1 / n, {n, 2, Infinity}]]
```

— `Table::iterb: Iterator {n, 2, ∞} does not have appropriate bounds.` >>

— `Table::iterb: Iterator {n, 2, ∞} does not have appropriate bounds.` >>

$$\text{Table}\left[\frac{1}{n}, \{n, 2, \infty\}\right]$$

Supremum nebo infimum nekonečné podmnožiny reálných čísel lze na počítači zjistit jen tehdy, pokud dovedeme onu množinu pro počítač vhodně popsat. Většinou to bývají hodnoty funkce např. na intervalu nebo na přirozených číslech.

V programu Mathematica bylo supremum a infimum nekonečně mnoha hodnot zavedeno až ve vyšších verzích programu. Používá se `MaxValue`, `MinValue` nebo `Maximize`, `Minimize` (poslední dva příkazy uvádějí i

bod, ve kterém se hodnoty dosahuje). Ne vždy však program hodnotu spočítá tak, jak očekáváme (viz 2. a 4. následující příklad).

```
MaxValue[{1/n, 2 ≤ n}, n]
```

$$\frac{1}{2}$$

```
MaxValue[{1/n, 2 < n}, n]
```

$$\frac{1}{2}$$

```
Maximize[{1/n, 2 ≤ n}, n]
```

$$\left\{\frac{1}{2}, \{n \rightarrow 2\}\right\}$$

```
Maximize[{1/n, 2 < n}, n]
```

— *Maximize::wksol:*

Warning: There is no maximum in the region in which the objective function is defined and the constraints are satisfied; Mathematica will return a result on the boundary. >>

$$\left\{\frac{1}{2}, \{n \rightarrow 2\}\right\}$$

Předchozí příklady jsou z hlediska počítače spočítány dobře, protože program chápe n v příkazu jako reálné číslo a my jsme měli na paměti přirozené číslo. To lze napravit přidáním podmínky, ale jen pro konečný počet čísel. Pro nekonečný počet je vhodné počítat výsledek numericky. Podívejte se však na špatný výsledek v předposledním příkladu.

```
MaxValue[{1/n, 2 < n < 100 && Element[n, Integers]}, n]
```

$$\frac{1}{3}$$

```
MaxValue[{1/n, n > 2 && Element[n, Integers]}, n]
```

$$\text{MaxValue}\left[\left\{\frac{1}{n}, n > 2 \wedge n \in \mathbb{Z}\right\}, n\right]$$

```
NMaxValue[{1/n, n > 2 && Element[n, Integers]}, n]
```

```
0.333333
```

```
Maximize[{1/n, 2 < n && Element[n, Integers]}, n]
```

$$\text{Maximize}\left[\left\{\frac{1}{n}, 2 < n \wedge n \in \mathbb{Z}\right\}, n\right]$$

```
NMaximize[{1/n, 2 < n && Element[n, Integers]}, n]
```

```
{0.333333, {n -> 3}}
```

```
Maximize[{1 - 1/n, 2 < n && Element[n, Integers]}, n]
```

$$\text{Maximize}\left[\left\{1 - \frac{1}{n}, 2 < n \wedge n \in \mathbb{Z}\right\}, n\right]$$

```
NMaximize[{1 - 1/n, 2 < n && Element[n, Integers]}, n]
```

```
{0.909091, {n -> 11}}
```

```
Maximize[{1 - 1/n, 2 < n}, n]
```

— *Maximize::natt*:

The maximum is not attained at any point satisfying the given constraints. >>

```
{1, {n -> ∞}}
```